
Introducci�n
En esta primera lección veremos los tipos de datos que .NET Framework pone a
nuestra disposición y cómo tendremos que usarlos desde Visual Basic 2008.

A continuación daremos un repaso a conceptos básicos o elementales sobre los
tipos de datos, que si bien nos serán familiares, es importante que lo veamos para
poder comprender mejor cómo están definidos y organizados los tipos de datos en
.NET.

Tipos de datos de .NET
Visual Basic 2008 está totalmente integrado con .NETFramework, por lo tanto, los
tipos de datos que podremos usar con este lenguaje serán los definidos en este
"marco de trabajo", por este motivo vamos a empezar usando algunas de las
definiciones que nos encontraremos al recorrer la documentación que acompaña a
este lenguaje de programación.

Los tipos de datos que podemos usar en Visual Basic 2008 son los mismo tipos de
datos definidos en .NET Framework y por tanto están soportados por todos los
lenguajes que usan esta tecnología. Estostipos comunes se conocen como el
Common Type System, (CTS), que traducido viene a significar el sistema de tipos
comunes de .NET. El hecho de que los tipos de datos usados en todos los
lenguajes .NET estén definidos por el propio Framework nos asegura que
independientementedel lenguaje que estemos usando, siempre utilizaremos el
mismo tipo interno de .NET, si bien cada lenguaje puede usar un nombre (o alias)
para referirse a ellos, aunque lo importante es que siempre serán los mismos
datos, independientemente de cómo se llame en cada lenguaje. Esto es una gran
ventaja, ya que nos permite usarlos sin ningún tipo de problemas para acceder a
ensamblados creados con otros lenguajes, siempre que esos lenguajes sean
compatibles con los tipos de datos de .NET.

En los siguientes enlaces tenemos los temas a tratar en esta primera lección del
módulo sobre las características del lenguaje Visual Basic 2008.

Tipos primitivos
Veamos en la siguiente tabla los tipos de datos definidos en .NET Framework y los
alias utilizados en Visual Basic 2008.

.NETFramework VB 2008
System.Boolean Boolean
System.Byte Byte
System.Int16 Short
System.Int32 Integer
System.Int64 Long
System.Single Single
System.Double Double
System.Decimal Decimal
System.Char Char
System.String String
System.Object Object
System.DateTime Date
System.SByte SByte
System.UInt16 UShort
System.UInt32 UInteger
System.UInt64 ULong

Tabla 2.1. Tipos de datos y equivalencia entre lenguajes

Debemos tener en cuenta, al menos si el rendimiento es una de nuestra
prioridades, que las cadenas en .NET son inmutables, es decir, una vez que se han
creado no se pueden modificar y en caso de que queramos cambiar el contenido,
.NET se encarga de desechar la anterior y crear una nueva cadena, por tanto si
usamos las cadenas para realizar concatenaciones (unión de cadenas para crear
una nueva), el rendimiento será muy bajo, si bien existe una clase en .NET que es



ideal para estos casos y cuyo rendimiento es superior al tipoString: la clase
StringBuilder.

Las últimas filas mostradas en la tabla son tipos especiales que si bien son parte
del sistema de tipos comunes (CTS) no formanparte de laCommon Language
Specification(CLS), es decir la especificación común para los lenguajes
"compatibles" con .NET, por tanto, si queremos crear aplicaciones que puedan
interoperar con todos los lenguajes de .NET, esos tipos no debemos usarlos como
valores de devolución de funciones ni como tipo de datos usado en los argumentos
de las funciones, propiedades o procedimientos.

Los tipos mostrados en la tabla 2.1 son los tipos primitivos de .NET y por
extensión de Visual Basic 2008, es decir son tipos "elementales" para los cuales
cada lenguaje define su propia palabra clave equivalente con el tipo definido en el
CTS de .NET Framework. Todos estos tipos primitivos podemos usarlos tanto por
medio de los tipos propios de Visual Basic, los tipos definidos en .NET o bien como
literales. Por ejemplo, podemos definir un número entero literal indicándolo con el
sufijoI:12345Io bien asignándolo a un valor de tipoIntegero a un tipo
Sytem.Int32de .NET. La única excepción de los tipos mostrados en la tabla1 es el
tipo de datosObject, este es un caso especial del que nos ocuparemos en la
próxima lección.

Sufijos o caracteres y símbolos identificadores para los tipos

Cuando usamos valores literales numéricos en Visual Basic 2008, el tipo de datos
que le asigna el compilador es el tipoDouble, por tanto si nuestra intención es
indicar un tipo de datos diferente podemos indicarlos añadiendo una letra como
sufijo al tipo, esto es algo que los más veteranos de VB6 ya estarán
acostumbrados, e incluso los más noveles también, en Visual Basic 2008 algunos
de ellos se siguen usando, pero el tipo asociado es el equivalente al de este nuevo
lenguaje (tal como se muestra en la tabla 1), por ejemplo para indicar un valor
entero podemos usar la letraIo el signo% , de igual forma, un valor de tipo
entero largo (Long) lo podemos indicar usandoL o &, en la siguiente tabla
podemos ver los caracteres o letra que podemos usar como sufijo en un literal
numérico para que el compilador lo identifique sin ningún lugar a dudas.

Tipo de datosSímboloCarácter
Short N.A. S
Integer % I
Long & L
Single ! F
Double # R
Decimal @ D
UShort N.A. US
UInteger N.A. UI
ULong N.A. UL

Tabla 2.2. Sufijos para identificar los tipos de datos

El uso de estos caracteres nos puede resultar de utilidad particularmente para los
tipos de datos que no se pueden convertir en un valor doble.



Nota:
Los sufijos pueden indicarse en minúsculas, mayúsculas o cualquier
combinación de mayúscula y minúscula.
Por ejemplo, el sufijo de un tipo ULongpuede ser: UL, Ul, ul, uL, LU,
Lu, lU o lu.
Para evitar confusiones, se recomienda siempre indicarlos en
mayúsculas, independientemente de que Visual Basic no haga ese
tipo de distinción.

Por ejemplo, si queremos asignar este valor literal a un tipoDecimal:
12345678901234567890, tal como vemos en la figura 1, el IDE de Visual Basic
2008 nos indicará que existe un error de desbordamiento (Overflow) ya que esa
cifra es muy grande para usarlo como valorDouble, pero si le agregamos el sufijo
D o@ ya no habrádudas de que estamos tratando con un valorDecimal.

Figura 2.1. Error de desbordamiento al intentar asignar un valor Double a
una variable Decimal

Tipos por valor y tipos por referencia

Los tipos de datos de .NET los podemos definir en dos grupos:

 Tipos por valor
 Tipos por referencia

Los tipos por valor son tipos de datos cuyo valor se almacena en la pila o en la
memoria "cercana", como losnuméricos que hemos visto. Podemos decir que el
acceso al valor contenido en uno de estos tipos es directo, es decir se almacena
directamente en la memoria reservada para ese tipo y cualquier cambio que
hagamos lo haremos directamente sobre dicho valor, deigual forma cuando
copiamos valores de un tipo por valor a otro, estaremos haciendo copias
independientes.

Por otro lado, los tipos por referencia se almacenan en el "monto" (heap) o
memoria "lejana", a diferencia de los tipos por valor, los tipos por referencia lo
único que almacenan es una referencia (o puntero) al valor asignado. Si hacemos
copias de tipos por referencia, realmente lo que copiamos es la referencia
propiamente dicha, pero no el contenido.

Estos dos casos los veremos en breve con más detalle.



Inferencia de tipos

Una de las características nuevas en Visual Basic 2008 es la inferencia de tipos.

Se conoce como inferencia de tipos a la característica de Visual Basic para inferir el
tipo de un dato al ser inicializado.

Para que la inferencia detipos sea efectiva, deberemos activar la opciónOption
Infera True, aunque por defecto, ese es el valor que tiene el compilador de
Visual Basic. Sin embargo, si se hace una migración de una aplicación de Visual
Basic a Visual Basic 2008, el valor de estaopción seráFalse.

Supongamos por lo tanto, la siguiente declaración:

Dim datoDeclarado = 2008

En este ejemplo, la variabledatoDeclarado, será una variable de tipoInteger
(Int32).

Si deseamos cambiar el tipo de dato a un tipoInt64por ejemplo, el compilador
nos devolverá un error. Así, el siguiente ejemplonoserá válido en Visual Basic
2008 con la opción de inferencia activada:

Dim datoDeclarado = 2008

datoDeclarado = Int64.MaxValue

Ahora bien, si cambiamos el valor deOption InferaFalse,el mismo ejemplo
será correcto.

¿Dónde está la diferencia?.
En este último caso, el caso de tener desactivada la opción de inferencia, la
declaración de la variabledatoDeclaradonos indica que es un tipo de datoObject
en su origen, y que al darle un valorInteger, ésta funciona como una variable
entera. Al cambiar su valor aLong, esta variable que es de tipoObject, cambia
sin problemas a valorLong.
En todo este proceso, hay un problema claro de rendimiento con acciones de
boxing y unboxing que no serían necesarias si tipáramos la variable con un tipo
concreto.
Eso es justamente lo que hace la opciónOption Inferpor nosotros. Nos permite
declarar una variable con el tipo inferido, y ese tipo de datos se mantiene dentro
de la aplicación, por lo que nosda la seguridad de que ese es su tipo de dato, y
que ese tipo de dato no va a variar.

Ver vídeo 1 de esta lección(Tiposde datos primitivos-video en
Visual Studio 2005 válido para Visual Studio 2008)

Ver vídeo 2 de esta lección(Tipos por valor y por referencia-video
en Visual Studio 2005 válido para Visual Studio 2008)



Variables y constantes
Disponer de todos estos tipos de datos no tendría ningún sentido si no pudiéramos
usarlos de alguna otra forma que de forma literal. Y aquí es donde entran en juego
las variables y constantes, no vamos a contarte qué son y para que sirven, salvo
en el caso de las constantes, ya que no todos los desarrolladores lasutilizamos de
la forma adecuada.

Consejo para usar las constantes

Siempre que tengamos que indicar un valor constante, ya sea para indicar el
máximo o mínimo permitido en un rango de valores o para comprobar el término
de un bucle, deberíamos usar una constante en lugar de un valor literal, de esta
forma si ese valor lo usamos en varias partes de nuestro código, si en un futuro
decidimos que dicho valor debe ser diferente, nos resultará más fácil realizar un
solo cambio que cambiarlo en todos los sitios enlos que lo hemos usado, además
de que de esta forma nos aseguramos de que el cambio se realiza adecuadamente
y no tendremos que preocuparnos de las consecuencias derivadas de no haber
hecho el cambio en todos los sitios que deberíamos.

Las constantes se definen utilizando la instrucciónConstseguida del nombre,
opcionalmente podemos indicar el tipo de datos y por último una asignación con el
valor que tendrá. Como veremos en la siguiente sección, podemos obligar a Visual
Basic 2008 a que en todas las constantes (y variables) que declaremos, tengamos
que indicar el tipo de datos.
Para declarar una constante lo haremos de la siguiente forma:

Const maximo As Integer = 12345678

Declarar variables

La declaración de las variables en Visual Basic 2008 se hace pormedio de la
instrucciónDimseguida del nombre de la constante y del tipo de datos que esta



contendrá. Con una misma instrucciónDimpodemos declarar más de una variable,
incluso de tipos diferentes, tal y como veremos a continuación.

La siguiente línea decódigo declara una variable de tipo entero:

Dim i As Integer

Tal y como hemos comentado, también podemos declarar en una misma línea más
de una variable:

Dim a, b, c As Integer

En este caso, las tres variables las estamos definiendo del mismo tipo, que esel
indicado al final de la declaración.

Nota:
Como hemos comentado, en Visual Basic 2008 se pueden declarar las
constantes y variables sin necesidad de indicar el tipo de datos que
contendrán, pero debido a que eso no es una buena práctica, a lo
largo deeste curso siempre declararemos las variables y constantes
con el tipo de datos adecuado a su uso.

Declarar variables y asignar el valor inicial

En Visual Basic 2008 también podemos inicializar una variable con un valor
distinto al predeterminado, que en los tipos numéricos es un cero, en las fechas es
el 1 de enero del año 1 a las doce de la madrugada (#01/01/0001
12:00:00AM#) y en la cadenas es un valor nulo (Nothing), para hacerlo,
simplemente tenemos que indicar ese valor, tal como veremos es muy parecido a
como se declaran las constantes. Por ejemplo:

Dim a As Integer = 10

En esa misma línea podemos declarar y asignar más variables, pero todas deben
estar indicadas con el tipo de datos:

Dim a As Integer = 10, b As Integer = 25

Por supuesto, el tipo de datos puede ser cualquiera de los tipos primitivos:

Dim a As Integer = 10, b As Integer = 25, s As String = "Hola"

Aunque para que el código sea más legible, y fácil de depurar, no deberíamos
mezclar en una misma instrucciónDimmás de un tipo de datos.

Nota:
Es importante saber que en las cadenas de Visual Basic 2008 el valor
de una variable de tipo String no inicializada NO es una cadena vacía,
sino un valor nulo (Nothing).



El tipo de datos Char

En Visual Basic 2008 podemos declarar valores de tipoChar, este tipo de datos es
un carácter Unicode y podemos declararlo y asignarlo a un mismo tiempo. El
problema con el que nos podemos encontrar es a la hora de indicar un carácter
literal.
Podemos convertir un valor numérico en un carácter o bien podemos convertir un
carácter en su correspondiente valor numérico.

Dim c As Char

c = Chr(65)

Dim n As Byte

n = Asc(c)

En Visual Basic 2008 los tiposCharse pueden asignar a variables de tipoStringy
se hará una conversión automática sin necesidad de utilizar funcionesde
conversión.

Si nuestra intención es asignar un valorChara una variable, además de la función
Chr, podemos hacerlo con un literal, ese valor literal estará encerrado entre
comillas dobles, (al igual que una cadena), aunque para que realmente sea un
carácter debemos agregarle unacjusto después del cierre de las comillas dobles:

Dim c As Char = "A"c

Obligar a declarar las variables con el tipo de datos

Visual Basic 2008 nos permite, (lamentablemente de forma predeterminada),
utilizar las variables y constantes sin necesidad de indicar el tipo de datos de
estas, pero, como comentábamos al principio, podemos obligar a que nos avise
cuando no lo estamos haciendo, ya que como decíamos en la nota, es una buena
costumbre indicarsiempreel tipo de datos que tendrán nuestras variables y
constantes.
Esa obligatoriedad la podemos aplicar a todo el proyecto o a un módulo en
particular, para ello tenemos que usar la instrucciónOption Strict On, una vez
indicado, se aplicará a todo el código, no solo a las declaraciones de variables,
constantes o al tipo de datos devuelto por las funciones y propiedades, sino
también a las conversiones y asignaciones entre diferentes tipos de datos.

No debemos confundirOption StrictconOption Explicit, este último, sirve para que
siempre tengamos que declarar todas las variables, mientras que el primero lo que
hace es obligarnos a que esas declaraciones tengan un tipo de datos.

Tanto una como la otra tienen dos estados: conectado o desconectado
dependiendo de que agreguemosOnuOffrespectivamente.

Insistimos en la recomendación de que siempre debemos "conectar" estas dos
opciones, si bienOption Explicit Onya viene como valor por defecto, cosa que no
ocurre conOption Strict, que por defecto está desconectado.



Aplicar Option Strict On a un ficheroen particular

Cuando agregábamos un nuevo fichero a nuestro proyecto de Visual Basic 2008 si
ya tenemos predefinida las opciones "estrictas", como es el caso deOption Explicit
On, estas no se añadirán a dicho fichero, (en un momento veremos cómo hacerlo
para que siempre estén predefinidas), pero eso no significa que no se aplique,
aunque siempre podemos escribir esas instrucciones (con el valorOnal final) en
cada uno de los ficheros de código que agreguemos a nuestro proyecto. Si nos
decidimos a añadirlas a los ficheros, esas líneas de código deben aparecer al
principio del fichero y solamente pueden estar precedidas de comentarios.

En la figura 2.1 mostrada en la lección anterior, tenemos una captura del editor de
Visual Basic 2008 enla que hemos indicado que queremos tener comprobación
estricta.

Aplicar Option Stict On a todo el proyecto

También podemos hacer queOption Strictfuncione igual queOption Explicit, es
decir, que esté activado a todo el proyecto, en este caso no tendríamos que
indicarlo en cada uno de los ficheros de código que formen parte de nuestro
proyecto, si bien solamente será aplicable a los que no tengan esas instrucciones,
aclaremos esto último: siOption Strict(uOption Explicit) está definido de forma
globalal proyecto, podemos desactivarlo en cualquiera de los ficheros, para ello
simplemente habría que usar esas declaraciones pero usandoOffen lugar deOn.
De igual forma, si ya está definido globalmente y lo indicamos expresamente, no
se producirá ningún error. Lo importante aquí es saber que siempre se usará el
estado indicado en cada fichero, independientemente de cómo lo tengamos
definido a nivel de proyecto.

Para que siempre se usen estas asignaciones en todo el proyecto, vamos a ver
cómo indicarlo en elentorno de Visual Basic 2008.

Abrimos Visual Studio 2008 y una vez que se haya cargado, (no hace falta crear
ningún nuevo proyecto, de este detalle nos ocuparemos en breve), seleccionamos
la opciónHerramientas>Opciones...se mostrará un cuadro de diálogoy del
panel izquierdo seleccionamos la opciónProyectos y soluciones, la expandimos
y seleccionamosValores predeterminados de VBy veremos ciertas opciones,
tal como podemos comprobar en la figura 2.2:



Figura 2.2. Opciones de proyectos (opciones mínimas)

De la lista despegableOption Strict, seleccionamosOn.Por defecto ya estarán
seleccionadas las opcionesOndeOption ExplicityBinarydeOption Compare,
por tanto no es necesario realizar ningún cambio más, para aceptar los cambios y
cerrar el cuadro de diálogo, presionamos en el botónAceptar.

Si en la ventana de opciones no aparece toda la configuración podemos hacer que
se muestren todas las disponibles. Para hacerlo, debemos marcar la casilla que
está en la parte inferior izquierda en la que podemos leer:Mostrar todas las
configuraciones, al seleccionar esa opción nos mostrará un número mayor de
opciones,tal como podemos ver en la figura 2.3:

Figura 2.3. Opciones de proyectos (todas las opciones)



Desde estemomento el compilador de Visual Basic se volverá estricto en todo lo
relacionado a las declaraciones de variables y conversiones, tal como vemos en la
figura 2.4 al intentar declarar una variable sin indicar el tipo de datos.

Figura 2.4. Aviso de Option Strict al declarar una variable sin tipo

Nota:
Una de las ventajas del IDE (Integrated Development Environment,
entorno de desarrollo integrado) de Visual Basic 2008 es que nos
avisa al momento de cualquier fallo que cometamos al escribir el
código, este "pequeño" detalle, aunque alguna veces puede llegar a
parecer fastidioso, nos facilita la escritura de código, ya que no
tenemos que esperar a realizar la compilación para que tengamos
constancia de esos fallos.

Más opciones aplicables a los proyectos

Aunque en estos módulos no trataremos a fondo el entorno de desarrollo, ya que
la finalidad de este curso online es tratar más en el código propiamente dicho,
vamos a mostrar otro de los sitios en los que podemos indicar dónde indicar que
se haga una comprobación estricta de tipos y, como veremos, también podremos
indicar algunas "nuevas peculiaridades" de Visual Basic 2008, todas ellas
relacionadas con el tema que estamos tratando.

Cuando tengamos un proyecto cargado en el IDE de Visual Studio 2008, (pronto
veremos cómo crear uno), podemos mostrar las propiedades del proyecto, para
ello seleccionaremos del menúProyectola opciónPropiedades de
<NombreDelProyecto>y tendremos un cuadro de diálogo como el mostrado en
la figura 2.5.



Figura 2.5. Ficha Compilar de las opciones del proyecto actual

Seleccionando la fichaCompilar, además de las típicas opciones deOption
Strict,Option Explicit,Option Comparey Option Infer, (estas asignaciones
solo serán efectivas para el proyecto actual), tendremos cómo queremos que
reaccione el compilador si se cumple algunas de las condiciones indicadas. Entre
esas condiciones, tenemos algo que muchos desarrolladores de Visual Basic
siempre hemos queridotener: Que nos avise cuando una variable la hemos
declarado pero no la utilizamos (Variable local no utilizada). Al tener marcada
esta opción (normalmente como unaAdvertencia), si hemos declarado una
variable y no la usamos en el código, (siempre que no le hayamos asignado un
valor al declararla), nos avisará, tal como podemos ver en la figura 2.6:

Figura 2.6. Aviso de variable no usada



Tipos anulables

Otra interesantísima característica de Visual Basic 2008 que conviene conocer, es
lo que se denominan tipos anulables.

Los tipos anulables no son nuevos en Visual Basic, de hecho su origen lo
encontramos en Visual Studio 2005, aunque eso sí, implementando la clase
Nullable(Of T).

Con Visual Basic 2008 no es necesario declarar ninguna clase para implementar
tipos anulables dentro de nuestras aplicaciones, y podemos declararlos de forma
directa.

Un tipo de dato anulable nos permitirá declarar una variable que podrá tener un
tipo de dato nulo.

Si hemos estado atentos hasta ahora, hemos podido ver que las variables
numéricas por ejemplo, se inicializan a 0.
Si quisiéramos por la razón que fuera, declarar esa variable como nula para saber
si en un determinado momento ha cambiado de valor o cualquier otra acción,
deberíamos utilizar los tipos de datos anulables, o bien, utilizar técnicas más
rudimentarias como una variable de tipoBooleanque nos permitiera saber si ha
habido un cambio de valor en una variable, sin embargo, coincidirá conmigo en
que el uso de un tipo de datos anulable es más natural y directo.

De esta forma, podríamos trabajar con tipos de datos anulables o que los
declararemos como nulos.
A continuación veremos un ejemplo:

Dim valor As Integer?

Para acceder al valor de un tipo anulable, podríamos hacerlo de la forma habitual,
ahora bien, si no sabemos si el valor es nulo o no, podríamos acceder a su valor
preguntando por él mediante la propiedadHasValue. La propiedadValuenos
indicará también, el valor de esa variable.
Un ejemplo que aclare esta explicación es el que podemos ver a continuación:

Dim valor As Integer?

If Valor.HasValue Then

MessageBox.Show(valor.Value)

End If

Otra característica de los tipos anulables es la posibilidad de utilizar la función
GetValueOrDefault.
Esta función nos permitirá acceder al valor de la variable si no es nulo, y al valor
que le indiquemos si es nulo.

Un breve ejemplo de este uso es el que se indica a continuación:

Dim valor As Integer?



valor = 2008

MessageBox.Show(valor.GetValueOrDefault(2012))

End If

En este ejemplo, el compilador nos devolvería el valor2008, ya que
GetValueOrDefaultsabe que la variable no posee un valor nulo y que por lo tanto,
debe obtener el valor no nulo de la variable anulable.
En el caso de que no hubiéramos dado ningún valor a la variable, la aplicación
obtendría el valor2012.

Tipos anónimos

Esta característica de Visual Basic 2008, nos permite declarar los tipos de datos de
forma implícita desde elcódigo de nuestras aplicaciones.

Un ejemplo práctico de declaración de tipos anónimos es el siguiente:

Dim declaracion = New With {.Nombre = "Carlos", .Edad = 27}

MessageBox.Show(String.Format("{0} tiene {1} años", _

declaracion.Nombre,
declaracion.Edad))

Como podemos ver en el ejemplo anterior, hemos declarado un objeto al que no
hemos indicado ningún tipo de dato concreto, pero a la hora de crear ese objeto,
hemos creado implícitamente un miembroNombrey un miembroEdad.

Ver vídeo 1 de esta lección(Declarar variables)-video en Visual
Studio 2005 válido para Visual Studio 2008

Ver vídeo 2 de esta lección(Definir constantes)-video en Visual
Studio 2005 válido para Visual Studio 2008



Enumeraciones: Constantes agrupadas
Una enumeración es una serie de constantes que están relacionadas entre sí. La
utilidad de las enumeraciones es más manifiesta cuando queremos manejar una
serie de valores constantes con nombre, es decir, podemos indicar un valor, pero
en lugar de usar un literal numérico, usamos un nombre, ese nombre es, al fin y al
cabo, una constante que tiene un valor numérico.

En Visual Basic 2008 las enumeraciones pueden ser de cualquier tipo numérico
integral, incluso enteros sin signo, aunque el valor predefinido es el tipoInteger.
Podemos declarar una enumeración de varias formas:

1-Sin indicar el tipo de datos, por tanto serán de tipoInteger:

Enum Colores

Rojo

Verde

Azul

End Enum

2-Concretando explícitamente el tipo de datos que realmente tendrá:

Enum Colores As Long

Rojo

Verde

Azul

End Enum



En este segundo caso, el valor máximo que podemos asignar a los miembros de
una enumeración será elque pueda contener un tipo de datosLong.

3-Indicando el atributoFlagsAttibute, (realmente no hace falta indicar el sufijo
Attributecuando usamos los atributos) de esta forma podremos usar los valores
de la enumeración para indicar valores que se pueden"sumar" o complementar
entre sí, pero sin perder el nombre, en breve veremos qué significa esto de "no
perder el nombre".

<Flags()> _

Enum Colores As Byte

Rojo = 1

Verde = 2

Azul = 4

End Enum

Nota:
Los atributos los veremos con más detalle enotra lección de este
mismo módulo.

El nombre de los miembros de las enumeraciones

Tanto si indicamos o no el atributoFlagsa una enumeración, la podemos usar de
esta forma:

Dim c As Colores = Colores.Azul Or Colores.Rojo

Es decir, podemos "sumar" los valores definidos en la enumeración. Antes de
explicar con detalle que beneficios nos puede traer el uso de este atributo, veamos
una característica de las enumeraciones.

Como hemos comentado, las enumeraciones son constantes con nombres, pero en
Visual Basic2008 esta definición llega más lejos, de hecho, podemos saber "el
nombre" de un valor de una enumeración, para ello tendremos que usar el método
ToString, (el cual se usa para convertir en una cadena cualquier valor numérico).
Por ejemplo, si tenemos la siguiente asignación:

Dim s As String = Colores.Azul.ToString

La variablescontendrá la palabra"Azul"no el valor 4.

Esto es aplicable a cualquier tipo de enumeración, se haya o no usado el atributo
FlagsAttribute.

Una vez aclarado este comportamiento de las enumeraciones en Visual Basic
2008, veamos que es lo que ocurre cuando sumamos valores de enumeraciones a
las que hemos aplicado el atributoFlagsy a las que no se lo hemos aplicado.
Empecemos por este último caso.

Si tenemos este código:



Enum Colores As Byte

Rojo = 1

Verde = 2

Azul = 4

End Enum

Dim c As Colores = Colores.Azul Or Colores.Rojo

Dim s As String = c.ToString

El contenido de la variablesserá"5", es decir, la representación numérica del
valor contenido:4 + 1, ya que el valorde la constanteAzules4 y el de la
constanteRojoes1.

Pero si ese mismo código lo usamos de esta forma (aplicando el atributoFlagsa la
enumeración):

<Flags()> _

Enum Colores As Byte

Rojo = 1

Verde = 2

Azul = 4

End Enum

Dim c As Colores = Colores.Azul Or Colores.Rojo

Dim s As String = c.ToString

El contenido de la variablesserá:"Rojo, Azul", es decir, se asignan los nombres
de los miembros de la enumeración que intervienen en ese valor, no el valor
"interno".

Los valores de una enumeración no son simples números

Como hemos comentado, los miembros de las enumeraciones realmente son
valores de un tipo de datos entero (en cualquiera de sus variedades) tal como
podemos comprobar en la figura 2.7:



Figura 2.7. Los tipos subyacentes posibles de una enumeración

Por tanto, podemos pensar que podemos usar cualquier valor para asignar a una
variable declarada como unaenumeración, al menos si ese valor está dentro del
rango adecuado.
En Visual Basic 2008 esto no es posible, al menos si lo hacemos de forma "directa"
y conOption Strictconectado, ya que recibiremos un error indicándonos que no
podemos convertir, por ejemplo, un valor entero en un valor del tipo de la
enumeración. En la figura 2.8 podemos ver ese error al intentar asignar el valor3
a una variable del tipoColores(definida con el tipo predeterminadoInteger).

Figura 2.8. Error al asignar un valor "normal" a una variable del tipo
Colores

El error nos indica que no podemos realizar esa asignación, pero el entorno
integrado de Visual Studio 2008 también nos ofrece alternativas para que ese
error no se produzca, esa ayuda se obtiene presionando en el signo de admiración
que tenemos justo donde está el cursor del mouse, pero no solo nos dice cómo
corregirlo, sino que también nos da la posibilidad de que el propio IDE se encargue
de corregirlo, tal como podemos apreciar en la figura 2.9.



Figura 2.9. Opciones de corrección de errores

Lo único que tendríamos que hacer es presionar en la sugerencia de corrección,
que en este caso es la única que hay, pero en otros casos pueden ser varias las
opciones y tendríamos que elegir la que creamos adecuada.

El código final (una vez corregido) quedaría de la siguiente forma:

Dim c As Colores = CType(3, Colores)

CTypees una de las formas que nos ofrece Visual Basic 2008 de hacer
conversiones entre diferentestipos de datos, en este caso convertimos un valor
entero en uno del tipoColores.

Si compilamos y ejecutamos la aplicación, ésta funcionará correctamente.
Aunque sabemos que es posible que usandoCTypeno asignemos un valor dentro
del rango permitido. En este caso, el valor 3 podríamos darlo por bueno, ya que es
la suma de 1 y 2 (RojoyVerde), pero ¿que pasaría si el valor asignado es, por
ejemplo, 15? En teoría no deberíamos permitirlo.

Estas validaciones podemos hacerlas de dos formas:
1-Con la clásicasolución de comprobar el valor indicado con todos los valores
posibles.
2-Usando funciones específicas del tipoEnum. Aunque en este último caso, solo
podremos comprobar los valores definidos en la enumeración.
En el siguiente ejemplo podemos hacer esa comprobación.

Sub mostrarColor(ByVal c As Colores)

' comprobar si el valor indicado es correcto

' si no está¡ definido, usar el valor Azul

If [Enum].IsDefined(GetType(Colores), c) = False Then

c = Colores.Azul

End If

Console.WriteLine("El color es {0}", c)

End Sub



Este código lo que hace es comprobar si el tipo de datosColorestiene definido el
valor contenido en la variablec, en caso de que no sea así, usamos un valor
predeterminado.

Nota:
La funciónIsDefinedsólo comprueba losvalores que se han definido
en la enumeración, no las posibles combinaciones que podemos
conseguir sumando cada uno de sus miembros, incluso aunque
hayamos usado el atributoFlagsAttribute.

Ver vídeo 1 de esta lección(Enumeraciones 1)-video en Visual
Studio 2005 válido para Visual Studio 2008

Ver vídeo 2 de esta lección(Enumeraciones 2)-video en Visual
Studio 2005 válido para Visual Studio 2008

Ver vídeo 3 de esta lección(Enumeraciones 3)-video en Visual
Studio 2005 válido para Visual Studio 2008

Ver vídeo 4 de esta lección(Enumeraciones 4)-video en Visual
Studio 2005 válido para Visual Studio 2008



Arrays (matrices)
Los arrays (o matrices) nos permitirán agrupar valores que de alguna forma
queremos que estén relacionados entre si.

Nota:
Esta es la definición usada en la documentación de Visual Studio
sobre qué es una matriz:
"Una matriz es una estructura de datos que contieneuna serie de
variables denominadas elementos de la matriz."
Aclaramos este punto, porque la traducción en castellano de Array
puede variar dependiendo del país, pero aquí utilizaremos la usada a
lo largo de la documentación de Visual Studio.

Declarar arrays

En C# los arrays se definen indicando un par de corchetes en el tipo de datos.

En Visual Basic 2008 la declaración de un array la haremos usando un par de
paréntesis en el nombre de la variable o del tipo, en el siguiente ejemplo
declaramos un array detipoStringllamadonombres:

Dim nombres() As String

Dim nombres As String()

Estas dos formas son equivalentes.

También podemos indicar el número de elementos que contendrá el array:

Dim nombres(10) As String



Pero solo podemos hacerlo en el nombre, si esacantidad de elementos lo
indicamos en el tipo, recibiremos un error indicándonos que"los límites de la
matriz no pueden aparecer en los especificadores del tipo".

Al declarar un array indicando el número de elementos, como es el caso anterior,
lo que estamos definiendo es un array de 11 elementos: desde cero hasta 10, ya
que en Visual Basic 2008, al igual que en el resto de lenguajes de .NET,todos los
arrays deben tener como índice inferior el valor cero.

Para que quede claro que el límite inferior debeser cero, en Visual Basic 2008
podemos usar la instrucción0 Topara indicar el valor máximo del índice superior,
ya que, tal como podemos comprobar si vemos 0 To 10, quedará claro que
nuestra intención es declarar un array con 11 elementos, o al menos nuestro
código resultará más legible:

Dim nombres(0 To 10) As String

Declarar e inicializar un array

En Visual Basic 2008 también podemos inicializar un array al declararlo, para ello
debemos poner los valores a asignar dentro de un para de llaves, tal como vemos
en el siguiente ejemplo:

Dim nombres() As String = {"Pepe", "Juan", "Luisa"}

Con el código anterior estamos creando un array de tipoStringcon tres valores
cuyos índices van de cero a dos.

En este caso, cuando iniciamos el array al declararlo, no debemos indicar el
número de elementos que tendrá ese array, ya que ese valor lo averiguará el
compilador cuando haga la asignación. Tampoco es válido indicar el número de
elementos que queremos que tenga y solo asignarle unos cuantos menos (o más),
ya que seproducirá un error en tiempo de compilación.

Si el array es bidimensional (o con más dimensiones), también podemos
inicializarlos al declararlo, pero en este caso debemos usar doble juego de llaves:

Dim nombres(,) As String = {{"Juan", "Pepe"}, {"Ana", "Eva"}}

En este código tendríamos un array bidimensional con los siguientes valores:

nombres(0,0)= Juan
nombres(0,1)= Pepe
nombres(1,0)= Ana
nombres(1,1)= Eva

Como podemos ver en la declaración anterior, si definimos arrays con más de una
dimensión, debemosindicarlas usando una coma para separar cada dimensión, o
lo que es más fácil de recordar: usando una coma menos del número de
dimensiones que tendrá el array. En los valores a asignar, usaremos las llaves
encerradas en otras llaves, según el número de dimensiones.



Aunque, la verdad, es que hay algunas veces hay que hacer un gran esfuerzo
mental para asociar los elementos con los índices que tendrán en el array, por
tanto, algunas veces puede que resulte más legible si indentamos o agrupamos
esas asignaciones, tal como vemos en el siguiente código:

Dim nomTri(,,) As String = _

{ _

{{"Juan", "Pepe"}, {"Luisa", "Eva"}}, _

{{"A", "B"}, {"C", "D"}} _

}

Console.WriteLine(nomTri(0, 0, 0)) ' Juan

Console.WriteLine(nomTri(0, 0, 1)) ' Pepe

Console.WriteLine(nomTri(0, 1, 0)) ' Luisa

Console.WriteLine(nomTri(0, 1, 1)) ' Eva

Console.WriteLine(nomTri(1, 0, 0)) ' A

Console.WriteLine(nomTri(1, 0, 1)) ' B

Console.WriteLine(nomTri(1, 1, 0)) ' C

Console.WriteLine(nomTri(1, 1, 1)) ' D

Tal como podemos comprobar, asi es más legible. Por suerte tenemos el carácter
del guión bajo para continuar líneas de código.

Cambiar el tamaño de un array

Para cambiar el tamaño de un array, usaremos la instrucciónReDim, esta
instrucción solo la podemos usar para cambiar el tamaño de un array previamente
declarado, no para declarar un array, ya que siempre hay que declarar
previamente los arrays antes de cambiarles el tamaño.

Dim nombres() As String

...

ReDim nombres(3)

nombres(0) = "Juan"

nombres(1) = "Pepe"

La mayor utilidad de esta instrucción, es que podemos cambiar el tamaño de un
array y mantener los valores que tuviera anteriormente, para lograrlo debemos
usarReDim Preserve.



ReDim Preserve nombres(3)

nombres(2) = "Ana"

nombres(3) = "Eva"

En este ejemplo, los valores que ya tuviera el arraynombres, se seguirían
manteniendo, y se asignarían los nuevos.

Si bien tantoReDimcomoReDim Preservese pueden usar en arrays de cualquier
número de dimensiones, en los arrays de más de una dimensión solamente
podemos cambiar el tamaño de la última dimensión.

Eliminar el contenido de un array

Una vez que hemos declarado un array y le hemos asignado valores, es posible
que nos interese eliminar esos valores de la memoria, para lograrlo, podemos
hacerlo de tres formas:

1.Redimensionando el array indicando que tiene cero elementos, aunque en el
mejor de los casos, si no estamos trabajando con arrays de más de una
dimensión, tendríamos un array de un elemento, ya que, como hemos
comentado anteriormente, los arrays de .NET el índice inferior es cero.

2.Usar la instrucciónErase. La instrucciónEraseelimina totalmente el array de
la memoria.

3.Asignar un valorNothingal array. Esto funciona en Visual Basic 2008 porque
los arrays realmente son tipos por referencia.

Los arrays son tipos por referencia

Como acabamos de ver, en Visual Basic 2008 los arrays son tipos por referencia, y
tal como comentamos anteriormente, los tipos por referencia realmente lo que
contienen son una referencia a los datos reales no los datos propiamente dichos.

¿Cual es el problema?

Veámoslo con un ejemplo y así lo tendremos más claro.

Dim nombres() As String = {"Juan", "Pepe", "Ana", "Eva"}

Dim otros() As String

otros = nombres

nombres(0) = "Antonio"



En este ejemplo definimos el arraynombresy le asignamos cuatro valores. A
continuación definimos otro array llamadootrosy le asignamos lo que tiene
nombres. Por último asignamos un nuevo valor al elemento cero del array
nombres.
Si mostramos el contenido de ambos arrays nos daremos cuenta de querealmente
solo existe una copia de los datos en la memoria, y tantonombres(0)como
otros(0)contienen el nombre"Antonio".

¿Qué ha ocurrido?

Que debido a que los arrays son tipos por referencia, solamente existe una copia
de los datos y tanto la variablenombres como la variable otros lo que contienen es
una referencia (o puntero) a los datos.

Si realmente queremos tener copias independientes, debemos hacer una copia del
array nombres en el array otros, esto es fácil de hacer si usamos el método
CopyTo. Éste método existe en todos los arrays y nos permite copiar un array en
otro empezando por el índice que indiquemos.
El único requisito es que el array de destino debe estar inicializado y tener espacio
suficiente para contener los elementos que queremos copiar.

En el siguiente código de ejemplo hacemos una copia del contenido del array
nombresen el arrayotros, de esta forma, el cambio realizado en el elemento
cero denombresno afecta al del arrayotros.

Dim nombres() As String = {"Juan", "Pepe", "Ana", "Eva"}

Dim otros() As String

ReDim otros(nombres.Length)

nombres.CopyTo(otros, 0)

nombres(0) = "Antonio"

Además del métodoCopyTo, los arrays tienen otros miembros que nos pueden ser
de utilidad, como por ejemplo la propiedadLengthusada en el ejemplopara saber
cuantos elementos tiene el arraynombres.

Para averiguar el número de elementos de un array, (realmente el índice
superior), podemos usar la funciónUBound.
También podemos usar la funciónLBound, (que sirve para averiguar el índice
inferior deun array), aunque no tiene ningún sentido en Visual Basic 2008, ya
que, como hemos comentado, todos los arrays siempre tienen un valor cero como
índice inferior.



Para finalizar este tema, solo nos queda por decir, que los arrays de Visual Basic
2008 realmente son tipos de datos derivados de la claseArrayy por tanto
disponen de todos los miembros definidos en esa clase, aunque de esto
hablaremos en la próxima lección, en la que también tendremos la oportunidad de
profundizar un poco más en los tipos porreferencia y en como podemos definir
nuestros propios tipos de datos, tanto por referencia como por valor.

Ver vídeo 1 deesta lección(Arrays Parte 1)-video en Visual Studio
2005 válido para Visual Studio 2008

Ver vídeo 2 de esta lección(Arrays Parte 2)-video en Visual Studio
2005 válido para Visual Studio 2008

Ver vídeo 3 de esta lección(Arrays Parte 3)-video en Visual Studio
2005 válido para Visual Studio 2008



Introducci�n
Enla lección anteriorvimos los tipos de datos predefinidos en .NET Framework, en
esta lección veremos cómo podemos crear nuestros propios tipos de datos, tanto
por valor como por referencia.
También tendremos ocasión de ver los distintos niveles de accesibilidad que
podemos aplicar a los tipos, así como a los distintos miembros de esos tipos de
datos. De los distintos miembros que podemos definir en nuestros tipos, nos
centraremos en las propiedades para ver en detalle los cambios que han sufrido
con respecto a VB6. También veremos temas relacionados con la programación
orientadaa objetos (POO) en general y de forma particular los que atañen a las
interfaces.

Clases y estructuras

 Clases: Tipos por referencia definidos por el usuario
o Las clases: El corazón de .NET Framework
o La herencia: Característica principal de la ProgramaciónOrientada a

Objetos
o Encapsulación y Polimorfismo
o Object: La clase base de todas las clases de .NET

 Definir una clase
o Una clase especial: Module
o Los miembros de una clase
o Características de los métodos y propiedades

 Accesibilidad, ámbito y miembros compartidos
 Parámetros y parámetros opcionales
 Array de parámetros opcionales (ParamArray)
 Sobrecarga de métodos y propiedades
 Parámetros por valor y parámetros por referencia

 Instanciar: Crear un objeto en la memoria
o Declarar primero la variable y después instanciarla
o Declarar y asignar en un solo paso



o El constructor: El punto de inicio de una clase
o Constructores parametrizados
o Cuando Visual Basic 2008 no crea un constructor automáticamente
o El destructor: El punto final de la vida de una clase
o Inicialización directa de objetos

 Estructuras: Tipos por valor definidos por el usuario
o Definir una estructura
o Constructores de las estructuras
o Destructores de las estructuras
o Los miembros de una estructura
o Cómo usar las estructuras

 Accesibilidad y «mbito
o Ámbito

 Ámbito de bloque
 Ámbito de procedimiento
 Ámbito de módulo
 Ámbito de espacio de nombres

o La palabra clave Global
o Accesibilidad

 Accesibilidad de las variables en los procedimientos
o Las accesibilidades predeterminadas
o Anidación de tipos

 Lostipos anidables
 El nombre completo de un tipo
 Importación de espacios de nombres
 Alias de espacios de nombres

 Propiedades
o Definir una propiedad
o Propiedades de solo lectura
o Propiedades de solo escritura
o Diferente accesibilidad para los bloques Get ySet
o Propiedades predeterminadas

 Sobrecarga de propiedades predeterminadas

 Interfaces
o ¿Qué es una interfaz?
o ¿Qué contiene una interfaz?
o Una interfaz es un contrato
o Las interfaces y el polimorfismo
o Usar una interfaz en una clase
o Acceder a los miembros implementados
o Saber si un objeto implementa una interfaz
o Implementación de múltiples interfaces
o Múltiple implementación de un mismo miembro
o ¿Dónde podemos implementar las interfaces?
o Un ejemplo práctico usando una interfaz de .NET



Clases: Tipos por referencia definidos por el usuario
Tal como vimos en la lección anterior, los tipos de datos se dividen en dos grupos:
tipos por valor y tipos por referencia. Los tipos por referencia realmente son
clases, de la cuales debemos crear una instancia para poder usarlas, esa instancia
o copia, se crea siempre en la memoria lejana (heap) y las variables lo único que
contienen es una referencia a la dirección de memoria en la que el CLR (Common
Language Runtime, motor en tiempo de ejecución de .NET), ha almacenado el
objeto recién creado.

En .NET Framework todo es de una forma u otra una clase, por tanto Visual Basic
2008 también depende de la creación de clases para su funcionamiento, ya que
todo el código que escribamos debemos hacerlo dentro de una clase.

Antes de entrar en detalles sintácticos, veamos la importancia que tienen las
clases en .NET Framework y como repercuten en las que podamos definir nosotros
usando VisualBasic 2008.

Las clases: el corazón de .NET Framework

Prácticamente todo lo que podemos hacer en .NET Framework lo hacemos
mediante clases. La librería de clases de .NET Framework es precisamente el
corazón del propio .NET, en esa librería de clases está todo lo que podemos hacer
dentro de este marco de programación; para prácticamente cualquier tarea que
queramos realizar existen clases, y si no existen, las podemos definir nosotros
mismos, bien ampliando la funcionalidad de alguna clase existente mediantela
herencia, bien implementando algún tipo de funcionalidad previamente definida o
simplemente creándolas desde cero.

La herencia: Caracter›stica principal de la Programaci�n Orientada a
Objetos

El concepto de Programación Orientada a Objetos (POO) es algointrínsico al
propio .NET Framework, por tanto es una característica que todos los lenguajes



basados en este "marco de trabajo" tienen de forma predeterminada, entre ellos el
Visual Basic 2008. De las características principales de la POO tenemos que
destacar la herencia, que en breve podemos definir como una característica que
nos permite ampliar la funcionalidad de una clase existente sin perder la que ya
tuviera previamente. Gracias a la herencia, podemos crear una nueva clase que se
derive de otra, esta nueva clase puede cambiar el comportamiento de la clase
base y/o ampliarlo, de esta forma podemos adaptar la clase, llamémosla, original
para adaptarla a nuestras necesidades.

El tipo de herencia que .NET Framework soporta es la herencia simple, es decir,
solo podemos usar una clase como base de la nueva, si bien, como veremos más
adelante, podemos agregar múltiple funcionalidad a nuestra nueva clase. Esta
funcionalidad nos servirá para aprovechar la funcionalidad de muchas de las clases
existentes en .NET Framework, funcionalidad que solamente podremos aplicar si
previamente hemos firmado un contrato que asegure a la clase de .NET que la
nuestra está preparada para soportar esa funcionalidad, esto lo veremos dentro de
poco con más detalle.

Encapsulaci�n y Polimorfismo

La encapsulación y el polimorfismo son otras dos características de la
programación orientada a objetos.

La encapsulación nos permite abstraer la forma que tiene de actuar una clase
sobre los datos que contiene o manipula, para poder lograrlose exponen como
parte de la clase los métodos y propiedades necesarios para que podamos manejar
esos datos sin tener que preocuparnos cómo se realiza dicha manipulación.

El polimorfismo es una característica que nos permite realizar ciertas acciones o
acceder a la información de los datos contenidos en una clase de forma semi-
anónima, al menos en el sentido de que no tenemos porqué saber sobre que tipo
objeto realizamos la acción, ya que lo único que nos debe preocupar es que
podemos hacerlo, por la sencilla razón de que estamos usando ciertos mecanismos
que siguen unas normas que están adoptadas por la clase.
El ejemplo clásico del polimorfismo es que si tengo un objeto que "sabe" cómo
morder, da igual que lo aplique a un ratón o a un dinosaurio, siempre ycuando
esas dos "clases" expongan un método que pueda realizar esa acción... y como
decía la documentación de Visual Basic 5.0, siempre será preferible que nos
muerda un ratón antes que un dinosaurio.

Object: La clase base de todas las clases de .NET

Todas las clases de .NET se derivan de la claseObject, es decir, lo indiquemos o
no, cualquier clase que definamos tendrá el comportamiento heredado de esa
clase. El uso de la claseObjectcomo base del resto de las clases de .NET es la
única excepción a la herencia simple soportada por .NET, ya que de forma
implícita, todas las clases de .NET se derivan de la claseObject
independientemente de que estén derivadas de cualquier otra.
Esta característica nos asegura que siempre podremos usar un objeto del tipo
Objectpara acceder a cualquier clase de .NET, aunque no debemos abrumarnos
todavía, ya que en el texto que sigue veremos con más detalle que significado
tiene esta afirmación.

De los miembros que tiene la claseObjectdebemos resaltar el métodoToString, el
cual ya lo vimos en la lección anterior cuando queríamos convertir un tipo



primitivo en una cadena. Este método está pensado para devolver una
representación en formato cadena de un objeto. El valor que obtengamos al usar
este método dependerá de cómo esté definido en cada clase y por defecto lo que
devuelve es el nombre completo de la clase, si bien en la mayoría de los casos el
valor que obtendremos al usar este método debería ser más intuitivo, por ejemplo
los tipos de datos primitivos tienen definidoeste método para devuelva el valor
que contienen, de igual forma, nuestras clases también deberían devolver un valor
adecuado al contenido almacenado. De cómo hacerlo, nos ocuparemos en breve.

Nota:
Todos los tipos de datos de .NET, ya sean por valor o porreferencia
siempre están derivados de la clase Object, por tanto podremos
llamar a cualquiera de los métodos que están definidos en esa clase.
Aunque en el caso de los tipos de datos por valor, cuando queremos
acceder a la clase Object que contienen, .NETFramework primero
debe convertirla en un objeto por referencia (boxing) y cuando
hemos dejado de usarla y queremos volver a asignar el dato a la
variable por valor, tiene que volver a hacer la conversión inversa
(unboxing).

Ver vídeo de esta lección(Clases 1)-video en Visual Studio 2005
válido para Visual Studio 2008



Definir una clase
En Visual Basic 2008, todo el código que queramos escribir, lo tendremos que
hacer en un fichero con la extensión.vb, dentro de ese fichero es donde
escribiremos nuestro código, el cual, tal comodijimos anteriormente siempre
estará incluido dentro de una clase, aunque un fichero de código de VB puede
contener una o más clases, es decir, no está limitado a una clase por fichero.

En Visual Basic 2008 las clases se definen usando la palabra claveClassseguida
del nombre de la clase, esa definición acaba indicándolo conEnd Class.

En el siguiente ejemplo definimos una clase llamada Cliente que tiene dos campos
públicos.

Class Cliente

Public Nombre As String

Public Apellidos As String

End Class

Una vez definida la clase podemos agregar los elementos (o miembros) que
creamos conveniente.
En el ejemplo anterior, para simplificar, hemos agregado dos campos públicos,
aunque también podríamos haber definido cualquiera de los miembros permitidos
en las clases.

Una clase especial: Module

En Visual Basic 2008 también podemos definir una clase especial llamadaModule,
este tipo de clase, como veremos, tiene un tratamiento especial.



La definición se hace usando la instrucciónModuleseguida del nombre a usar y
acaba con End Module.
Cualquier miembro definido en unModulesiempre estará accesible en todo el
proyecto y para usarlos no tendremos que crear ningún objeto en memoria.
Las clases definidas con la palabra claveModulerealmente equivalen a las clases
en las que todos los miembros están compartidos y por tanto siempre disponibles
a toda la aplicación.

De todos estos conceptos nos ocuparemos en las siguientes lecciones, pero es
necesario explicar que existe este tipo de clase ya que será el tipo de datos que el
IDE de Visual Basic 2008 usará al crear aplicaciones del tipo consola, ya que ese
será el tipo de proyecto que crearemos para practicar con el código mostrado en
este primer módulo.

Los miembros de una clase

Una clase puede contener cualquiera de estos elementos (miembros):

 Enumeraciones
 Campos
 Métodos (funciones o procedimientos)
 Propiedades
 Eventos

Las enumeraciones, como vimos en la lección anterior, podemos usarlas para
definir valores constantes relacionados, por ejemplo para indicar los valores
posibles de cualquier "característica" de la clase.

Los camposson variables usadas para mantener los datos que la clase
manipulará.

Los métodosson las acciones que la clase puede realizar, normalmente esas
acciones serán sobre los datos que contiene. Dependiendo de que el método
devuelva o no un valor, podemos usar métodos de tipoFunctiono de tipoSub
respectivamente.

Las propiedadesson las "características" de las clases y la forma de acceder
"públicamente" a los datos que contiene. Por ejemplo, podemos considerar que el
nombre y los apellidos de un cliente son dos características del cliente.

Los eventosson mensajes que la clase puede enviar para informar que algo está
ocurriendo en la clase.

Características de los métodos y propiedades

Accesibilidad, «mbito y miembros compartidos

Aunque estos temas los veremos en breve con más detalle, para poder
comprender mejor las características de los miembros de una clase (o cualquier
tipo que definamos), daremos un pequeño adelanto sobre estas características que
podemos aplicar a los elementos que definamos.

Accesibilidad y ámbito son dos conceptos que están estrechamente relacionados.
Aunque en la práctica tienen el mismo significado, ya que lo que representan es la



"cobertura" o alcance que tienenlos miembros de las clases e incluso de las
mismas clases que definamos.

Si bien cada uno de ellos tienen su propia "semántica", tal como podemos ver a
continuación:

Ámbito

Es el alcance que la definición de un miembro o tipo puede tener. Es decir, cómo
podemos acceder a ese elemento y desde dónde podemos accederlo.
El ámbito de un elemento de código está restringido por el "sitio" en el que lo
hemos declarado. Estossitiospueden ser:

 Ámbito de bloque:Disponible únicamente en el bloque de código en el que
se ha declarado.

 Ámbito de procedimiento:Disponible únicamente dentro del
procedimiento en el que se ha declarado.

 Ámbito de módulo:Disponible en todo el código del módulo, la clase o la
estructura donde se ha declarado.

 Ámbito de espacio de nombres:Disponible en todo el código del espacio
de nombres.

Accesibilidad

A los distintos elementos de nuestro código (ya sean clases o miembros de las
clases) podemos darle diferentes tipos de accesibilidad. Estos tipos de "acceso"
dependerán del ámbito que queramos que tengan, es decir, desde dónde
podremos accederlos.
Los modificadores de accesibilidad son:

 Public:Acceso no restringido.
 Protected:Acceso limitado a la clase contenedora o a los tipos derivados de

esta clase.
 Friend:Acceso limitado al proyecto actual.
 Protected Friend:Acceso limitado al proyecto actual o a los tipos derivados

de la clase contenedora.
 Private:Acceso limitado al tipo contenedor.

Por ejemplo, podemos declarar miembros privados a una clase, en ese caso,
dichos miembros solamente los podremos acceder desde la propia clase, pero no
desde fuera de ella.

Nota:
Al declarar una variable conDim, el ámbito que le estamos aplicando
por regla general, es privado, pero como veremos, en Visual Basic
2008 el ámbito puede ser diferente aprivado dependiendo del tipo en
el que declaremos esa variable.

Miembros compartidos

Por otro lado, los miembros compartidos de una clase o tipo, son elementos que
no pertenecen a una instancia o copia en memoria particular, sino que pertenecen
al propiotipo y por tanto siempre están accesibles o disponibles, dentro del nivel



del ámbito y accesibilidad que les hayamos aplicado, y sutiempo de vidaes el
mismo que el de la aplicación.

Del ámbito, la accesibilidad y los miembros compartidos nos ocuparemos con más
detalle en unalección posterior, donde veremos ciertas peculiaridades, como
puede ser la limitación del ámbito de un miembro queaparentementetiene una
accesibilidad no restringida.

Par«metros espec›ficos y par«metros opcionales

En Visual Basic 2008, tanto los miembros de una clase, (funciones y métodos
Sub), como las propiedades pueden recibir parámetros. Esos parámetros pueden
estar explícitamente declarados, de forma que podamos indicar cuantos
argumentos tendremos que pasar al método, también podemos declarar
parámetros opcionales, que son parámetros que no hace falta indicar al llamar a la
función o propiedad, y que siempre tendrán un valor predeterminado, el cual se
usará en caso de que no lo indiquemos.
Además de los parámetros específicos yopcionales, podemos usar un array de
parámetros opcionales, en los que se puede indicar un número variable de
argumentos al llamar a la función que los define, esto lo veremos en la siguiente
sección.

Nota: Sobre parámetros y argumentos
Para clarificar lascosas, queremos decir que los parámetros son los
definidos en la función, mientras que los argumentos son los
"parámetros" que pasamos a esa función cuando la utilizamos desde
nuestro código.

Con los parámetros opcionales lo que conseguimos es permitir que el usuario no
tenga que introducir todos los parámetros, sino solo los que realmente necesite,
del resto, (los no especificados), se usará el valor predeterminado que hayamos
indicado, porque una de las "restricciones" de este tipo de parámetros, es que
siempre debemos indicar también el valor por defecto que debemos usar en caso
de que no se especifique.

Veamos unos ejemplo para aclarar nuestras ideas:

Function Suma(n1 As Integer, Optional n2 As Integer = 15) As
Integer

Suma = n1 + n2

End Function

Eneste primer ejemplo, el primer parámetro es obligatorio (siempre debemos
indicarlo) y el segundo es opcional, si no se indica al llamar a esta función, se
usará el valor 15 que es el predeterminado.

Para llamar a esta función, lo podemos hacer de estas tres formas:

' 1- indicando los dos parámetros (el resultado será 4= 1 + 3)

t = Suma(1, 3)



' 2- Indicando solamente el primer parámetro (el resultado será
16= 1 + 15)

t = Suma(1)

' 3- Indicando los dos parámetros, pero en el opcional usamos el
nombre

t = Suma(1, n2:= 9)

El tercer ejemplo solamente tiene utilidad si hay más de un parámetro opcional.

Nota:
Los parámetros opcionales deben aparecer en la lista de parámetros
del método, después de los "obligatorios"

Nota:
En el caso hipotético y no recomendable deque no estemos usando
Option Strict On, tanto los parámetros normales como los opcionales
los podemos indicar sin el tipo de datos, pero si en alguno de ellos
especificamos el tipo, debemos hacerlo en todos.

Array de par«metros opcionales (ParamArray)

Encuanto al uso deParamArray, este tipo de parámetro opcional nos permite
indicar un número indeterminado de parámetros,en Visual Basic 2008siempre
debe ser una array e internamente se trata como un array normal y corriente, es
decir, dentro del método opropiedad se accede a él como si de un array se
tratara... entre otras cosas, ¡porque es un array!

Si tenemos activadoOption Strict, el array usado conParamArraydebe ser de un
tipo en concreto, si queremos que acepte cualquier tipo de datos, lo podemos
declarar con el tipoObject.

Si por cualquier razón necesitamos pasar otros valores que no sean de un tipo en
concreto, por ejemplo, además de valores "normales" queremos pasar un array,
podemos desactivarOption Strictpara hacerlo fácil.
Por ejemplo, siqueremos hacer algo como esto:

Function Suma(ParamArray n() As Object) As Integer

Dim total As Integer

'

For i As Integer = 0 To n.Length - 1

If IsArray(n(i)) Then

For j As Integer = 0 To n(i).Length - 1

total += n(i)(j)

Next

Else



total += n(i)

End If

Next

Return total

End Function

' Para usarlo:

Dim t As Integer

Dim a(2) As Integer = {1, 2, 3}

t = Suma(a, 4, 5, 6)

Console.WriteLine(t)

Tal como vemos, el primer argumento que pasamos a la función Suma es un
array, después pasamos tres valores enteros normales. El resultado de ejecutar
ese código será el valor21, como es de esperar.

Nota:
Como Visual Basic 2008 no sabe que tipo contiene n(i), al escribir
este código, no nos informará de que ese "objeto" tiene una
propiedad llamada Length.

Pero comonodebemos "acostumbrarnos" a desactivarOption Strict, vamos a ver
el código que tendríamos que usar para que también acepte un array como parte
de los argumentos indicados al llamar a la función.

Function Suma(ByVal ParamArray n() As Object) As Integer

Dim total As Integer

'

For i As Integer = 0 To n.Length - 1

If IsArray(n(i)) Then

For j As Integer = 0 To CType(n(i), Integer()).Length
- 1

total += CType(n(i), Integer())(j)

Next

Else

total += CInt(n(i))

End If



Next

Return total

End Function

' Para usarlo:

Dim t As Integer

Dim a(2) As Integer = {1, 2, 3}

t = Suma(a, 4, 5, 6)

Console.WriteLine(t)

Como podemos comprobar, al tenerOption Strictactivado, debemos hacer una
conversión explícita del array a uno de un tipo en concreto, como es natural, en
este código estamos suponiendo que el array pasado como argumento a la función
es de tipoInteger, en caso de que no lo fuera, recibiríamos un error en tiempo de
ejecución.

Para comprobar si el contenido de n(i) es un array, hemos usado la función
IsArraydefinida en el espacio de nombresMicrosoft.VisualBasic(que por defecto
utilizan todas las aplicaciones de Visual Basic 2008), pero también lo podríamos
haber hecho más al "estilo .NET", ya que, como sabemos todos los arrays de .NET
realmente se derivan de la claseArray:

If TypeOf n(i) Is Array Then

Nota:
Cuando queramos usar ParamArray para recibir un array de
parámetros opcionales, esta instrucción debe ser la última de la lista
de parámetros de la función (método).

Tampoco se permite tener parámetros opcionales y ParamArray en la
misma función.

Sobrecarga de métodos y propiedades

La sobrecarga de funciones (realmente de métodos y propiedades), es una
característica que nos permite tener una misma función con diferentes tipos de
parámetros, ya sea en número o en tipo.

Supongamos que queremos tener dos funciones (o más) que nos permitan hacer
operaciones con diferentes tipos de datos, y que, según el tipo de datos usado, el
valor que devuelva sea de ese mismo tipo.

En este ejemplo, tenemos dos funciones que se llaman igual pero una recibe
valores de tipo entero y la otra de tipo decimal:



Function Suma(n1 As Integer, n2 As Integer) As Integer

Return n1 + n2

End Function

Function Suma(n1 As Double, n2 As Double) As Double

Return n1 + n2

End Function

Como podemos comprobar las dos funciones tienen el mismo nombre, pero tanto
una como otra reciben parámetros de tipos diferentes.

Con Visual Basic 2008 podemos sobrecargar funciones, pero lo interesante no es
que podamos hacerlo, sino cómo podemos usar esas funciones. En el código
anterior tenemos dos funciones llamadasSuma, laprimera acepta dos parámetros
de tipoIntegery la segunda de tipoDouble. Lo interesante es que cuando
queramos usarlas, no tenemos que preocuparnos de cual vamos a usar, ya que
será el compilador el que decida la más adecuada al código que usemos, por
ejemplo:

' En este caso, se usará la que recibe dos valores enteros

Console.WriteLine(Suma(10, 22) )

' En este caso se usará la que recibe valores de tipo Double

Console.WriteLine(Suma(10.5, 22)

El compilador de Visual Basic 2008 es el que decide que función usar, esa decisión
la toma a partir de los tipos de parámetros que hayamos indicado. En el segundo
ejemplo de uso, el que mejor coincide es el de los dos parámetros de tipoDouble.

También podemos tener sobrecarga usando una cantidad diferente de parámetros,
aunque éstos sean del mismo tipo. Por ejemplo, podemos añadir esta declaración
al código anterior sin que exista ningún tipo de error, ya que esta nueva función
recibe tres parámetros en lugar de dos:

Function Suma(n1 As Integer, n2 As Integer, n3 As Integer) As
Integer

Return n1 + n2 + n3

End Function

Por tanto, cuando el compilador se encuentre con una llamada a la funciónSuma
en la que se usen tres parámetros, intentará usar esta última.



Nota:
Para que exista sobrecarga, la diferencia debe estar en el número o
en el tipo de los parámetros, no en el tipo del valor devuelto.

Cuando, desde el IDE de Visual Basic 2008, queremos usar los métodos
sobrecargados, nos mostrará una lista de opciones indicándonos las posibilidades
de sobrecarga que existeny entre las que podemos elegir, tal como vemos en la
figura 2.10:

Figura 2.10. Lista de parámetros soportados por un método

Cuando utilicemos parámetros opcionales debemos tener en cuenta que puede que
el compilador nos muestre un error, ya que es posible que esa función que declara
parámetros opcionales entre en conflicto con una "sobrecargada".

Por ejemplo:

Function Suma(n1 As Integer, Optional n2 As Integer = 33) As
Integer

Return n1 + n2

End Function

Si tenemos esta declaración además de las anteriores, el programa no compilará,
ya que si hacemos una llamada a la funciónSumacon dos parámetros enteros, el
compilador nosabrá si usar esta última o la primera que declaramos, por tanto
producirá un error.

Parámetros por valor y parámetros por referencia

Al igual que tenemos dos tipos de datos diferentes, en los parámetros de las
funciones también podemos tenerlos, para ello tendremos que usarByValoByRef
para indicar al compilador cómo debe tratar a los parámetros.

Cuando un parámetro es por valor (ByVal), elruntimeantes de llamar a la función
hace una copia de ese parámetro y pasa la copia a la función, por tanto cualquier
cambio que hagamos a ese parámetro dentro de la función no afectará al valor
usado "externamente".

En el caso de que el parámetro sea por referencia (ByRef), el compilador pasa una
referencia que apunta a la dirección de memoria en la que están los datos, por
tanto si realizamos cambios dentro de la función, ese cambio si que se verá
reflejado en el parámetro usado al llamar a la función.



Nota:
Hay que tener en cuenta que si pasamos un objeto a una función, da
igual que lo declaremos por valor o por referencia, ya que en ambos
casos se pasa una referencia a la dirección de memoria en la que
están los datos, porque, como sabemos, las variables de los tipos por
referencia siempre contienen una referencia a los datos, no los datos
en sí.

Cuando en Visual Basic 2008 usamos parámetros en los que no se indica si es por
valor (ByVal) o por referencia (ByRef), serán tratados como parámetros por valor.

Nota:
Si usamos el IDE de Visual Studio 2008 para escribir el código, no
debemos preocuparnos de este detalle, ya que si no indicamos si es
por valor o por referencia, automáticamente le añadirá la palabra
clave ByVal, para que no haya ningún tipo de dudas.

Ver vídeo 1 de esta lección(Clases 2: Los miembros de las clases)-
video en Visual Studio 2005 válido para Visual Studio 2008

Ver vídeo 2 de esta lección(Clases 3: Las propiedades)-video en
Visual Studio 2005 válido para Visual Studio 2008

Ver vídeo 3 de esta lección(Clases 4: Accesibilidad)-video en
Visual Studio 2005 válido para Visual Studio 2008



Instanciar: Crear un objeto en la memoria
Una vez que tenemos una clase definida, lo único de lo que disponemos es de una
especie de plantilla o molde a partir del cual podemos crear objetos en memoria.

Para crear esos objetos en Visual Basic 2008 lo podemos hacer de dos formas,
pero como veremos siempre será mediante la instrucciónNewque es la encargada
de crear el objeto en la memoria y asignar la dirección del mismo a la variable
usada en la parte izquierda de la asignación.

Declarar primero la variable y después instanciarla

Lo primero que tenemos que hacer es declarar una variable del tipo que queremos
instanciar, esto lo hacemos de la misma forma que con cualquier otro tipo de
datos:

Dim c As Cliente

Con esta línea de código lo que estamos indicando a Visual Basic es que tenemos
intención de usar una variable llamadacpara acceder a una clase de tipoCliente.
Esa variable, cuando llegue el momento de usarla, sabrá todo lo que hay que
saber sobre una claseCliente, pero hasta que no tenga una "referencia" a un
objeto deese tipo no podremos usarla.

La asignación de una referencia a un objetoClientela haremos usando la
instrucciónNewseguida del nombre de la clase:

c = New Cliente

A partir de este momento, la variablectiene acceso a un nuevo objeto del tipo
Cliente, por tanto podremos usarla para asignarle valores y usar cualquiera de los
miembros que ese tipo de datos contenga:



c.Nombre = "Antonio"

c.Apellidos = "Ruiz Rodríguez"

Declarar y asignar en un solo paso

Con las clases o tipos por referencia también podemos declarar una variable y al
mismo tiempo asignarle un nuevo objeto:

Dim c As New Cliente

O también:

Dim c As Cliente = New Cliente

Las dos formas producen el mismo resultado, por tanto es recomendable usar la
primera.

El constructor: El punto de inicio de una clase

Cada vez que creamos un nuevo objeto en memoria estamos llamando al
constructor de la clase. En Visual Basic 2008 el constructor es un método de tipo
SubllamadoNew.

En el constructor de una clase podemos incluir el código que creamos conveniente,
pero realmente solamente deberíamos incluir el que realice algún tipo de
inicialización, en caso de que no necesitemos realizar ningún tipo de inicialización,
no es necesario definir el constructor, ya que el propio compilador lo hará por
nosotros. Estoes así porque todas las clases deben implementar un constructor,
por tanto si nosotros no lo definimos, lo hará el compilador de Visual Basic 2008.

Si nuestra claseClientetiene un campo para almacenar la fecha de creación del
objeto podemos hacer algo como esto:

Class Cliente

Public Nombre As String

Public Apellidos As String

Public FechaCreacion As Date

'

Public Sub New()

FechaCreacion = Date.Now

End Sub

End Class

De esta forma podemos crear un nuevoClientey acto seguidocomprobar el valor
del campoFechaCreacionpara saber la fecha de creación del objeto.



En los constructores también podemos hacer las inicializaciones que, por ejemplo
permitan a la clase a conectarse con una base de datos, abrir un fichero o cargar
una imagen gráfica, etc.

Constructores parametrizados

De la misma forma que podemos tener métodos y propiedades sobrecargadas,
también podemos tener constructores sobrecargados, ya que debemos recordar
que en Visual Basic 2008, un constructor realmente es un método de tipoSub, y
como todos los métodos y propiedades de Visual Basic 2008, también admite la
sobrecarga.

La ventaja de tener constructores que admitan parámetros es que podemos crear
nuevos objetos indicando algún parámetro, por ejemplo un fichero a abrir o, en el
caso de la claseCliente, podemos indicar el nombre y apellidos del cliente o
cualquier otro dato que creamos conveniente.

Para comprobarlo, podemos ampliar la clase definida anteriormente para que
también acepte la creación de nuevos objetosindicando el nombre y los apellidos
del cliente.

Class Cliente

Public Nombre As String

Public Apellidos As String

Public FechaCreacion As Date

'

Public Sub New()

FechaCreacion = Date.Now

End Sub

'

Public Sub New(elNombre As String, losApellidos As String)

Nombre = elNombre

Apellidos = losApellidos

FechaCreacion = Date.Now

End Sub

End Class

Teniendo esta declaración de la claseCliente, podemos crear nuevos clientes de
dos formas:

Dim c1 As New Cliente



Dim c2 As New Cliente("Jose", "Sánchez")

Como podemos comprobar, en ciertos casos es más intuitiva la segunda forma de
crear objetos del tipoCliente, además de que así nos ahorramos de tener que
asignar individualmente los camposNombreyApellidos.

Esta declaración de la claseClientela podríamos haber hecho de una forma
diferente. Por un lado tenemos un constructor "normal" (no recibe parámetros) en
el que asignamos la fecha de creación y por otro el constructor que recibe los
datos del nombre y apellidos. En ese segundo constructor también asignamos la
fecha de creación, ya que, se instancie como se instancie la clase, nos interesa
saber siempre la fecha de creación. En este ejemplo, por su simpleza no es
realmente un problema repetir la asignación de la fecha, pero si en lugar de una
inicialización necesitáramos hacer varias, la verdad es que nos encontraríamos con
mucha "duplicidad" de código. Por tanto, en lugar de asignar los datos en dos
lugares diferentes, podemos hacer esto otro:

Class Cliente

Public Nombre As String

Public Apellidos As String

Public FechaCreacion As Date

'

Public Sub New()

FechaCreacion = Date.Now

End Sub

'

Public Sub New(elNombre As String, losApellidos As String)

Nombre = elNombre

Apellidos = losApellidos

End Sub

End Class

Es decir, desde el constructor con argumentos llamamos al constructor que no los
tiene, consiguiendo que también se asigne la fecha.

Esta declaración de la claseClienterealmente no compilará. Y no compila por la
razón tan "simple" de que aquí el compilador de Visual Basic 2008 no sabe cual es
nuestra intención, ya queNewes una palabra reservada que sirve para crear
nuevos objetos y, siempre, una instrucción tiene preferencia sobre el nombre de
un método, por tanto, podemos recurrir al objeto especialMeel cual nos sirve,
para representar al objeto que actualmente está en la memoria, así que, para que
esa declaración de la clase Cliente funcione, debemos usarMe.New()para llamar
al constructor sin parámetros.



Nota:
El IDE de Visual Basic 2008 colorea las instrucciones y tipos propios
del lenguaje, en el caso de Me, lo que no debe confundirnos es que
cuando declaramos Sub New, tanto Sub como New se muestran
coloreadas, cuando solo se debería colorear Sub; sin embargo
cuando usamos Me.New, solo se colorea Me y no New que es
correcto, tal como vemos en la figura 2.11, ya que en este caso New
es el nombre de un procedimiento y los procedimientos no son parte
de las instrucciones y tipos de .NET.

Figura 2.11. Coloreo erróneo de New

Cuando Visual Basic 2008 no crea un constructor autom«ticamente

Tal como hemos comentado, sinosotros no definimos un constructor (Sub New),
lo hará el propio compilador de Visual Basic 2008, y cuando lo hace
automáticamente, siempre es un constructor sin parámetros.

Pero hay ocasiones en las que nos puede interesar que no exista un constructor
sin parámetros, por ejemplo, podemos crear una claseClienteque solo se pueda
instanciar si le pasamos, por ejemplo el número de identificación fiscal, (NIF), en
caso de que no se indique ese dato, no podremos crear un nuevo objetoCliente,
de esta forma,nos aseguramos siempre de que el NIF siempre esté especificado.
Seguramente por ese motivo, si nosotros definimos un constructor con
parámetros, Visual Basic 2008 no crea uno automáticamente sin parámetros. Por
tanto, si definimos un constructor con parámetros en una clase y queremos que
también tenga uno sin parámetros, lo tenemos que definir nosotros mismos.

El destructor: El punto final de la vida de una clase

De la misma forma que una clase tiene su punto de entrada o momento de
nacimiento en el constructor, también tienen un sitio que se ejecutará cuando el
objeto creado en la memoria ya no sea necesario, es decir, cuando acabe la vida
del objeto creado.

El destructor de Visual Basic 2008 es un método llamadoFinalize, el cual se
hereda de la claseObject, por tanto no es necesario que escribamos nada en él. El
propio CLR se encargará de llamar a ese método cuando el objeto ya no sea
necesario.

La forma en que se destruyen los objetos pueden ser de dos formas:
Porque los destruyamos nosotros asignando unvalorNothinga la variable que lo
referencia, o bien porque el objeto esté fuera de ámbito, es decir, haya salido de



su espacio de ejecución, por ejemplo, si declaramos un objeto dentro de un
método, cuando ese método termina, el objeto se destruye, (hayalgunas
excepciones a esta última regla, como puede ser que ese mismo objeto también
esté referenciado por otra variable externa al método.)

Lo que debemos tener muy presente es que en .NET los objetos no se destruyen
inmediatamente. Esto es así debido aque en .NET existe un "sistema" que se
encarga de realizar esta gestión de limpieza: El recolector de basura o de objetos
no usados (Garbage Collector, GC). Este recolector de objetos no usados se
encarga de comprobar constantemente cuando un objeto no seestá usando y es el
que decide cuando hay que llamar al destructor.

Debido a esta característica de .NET, si nuestra clase hace uso de recursos
externos que necesiten ser eliminados cuando el objeto ya no se vaya a seguir
usando, debemos definir un métodoque sea el encargado de realizar esa
liberación, pero ese método debemos llamarlo de forma manual, ya que, aunque
en .NET existen formas de hacer que esa llamada sea automática, nunca
tenderemos la seguridad de que se llame en el momento oportuno, y esto es algo
que, según que casos, puede ser un inconveniente.

Recomendación:
Si nuestra clase utiliza recursos externos, por ejemplo un fichero o
una base de datos, debemos definir un método que se encargue de
liberarlos y a ese método debemos encargarnos de llamarlo cuando
ya no lo necesitemos.
Por definición a este tipo de métodos se les suele dar el nombre
Close o Dispose, aunque este último tiene un significado especial y
por convención solo debemos usarlo siguiendo las indicaciones de la
documentación.

Inicialización directa de objetos

De todos los modos, en Visual Basic 2008 se ha agregado una característica
adicional que tiene que ver con la inicialización de objetos.

Hasta ahora hemos visto como instanciar una clase utilizando su constructor, pero
quizáslo que no sabíamos es que en Visual Basic 2008 podemos inicializar la clase
a través de su constructor e incluso inicializar directamente las propiedades de la
clase.

Supongamos por lo tanto el siguiente código fuente:

Class Cliente

Private m_Nombre As String

Public Property Nombre() As String

Get

Return m_Nombre

End Get

Set(ByVal value As String)



m_Nombre = value

End Set

End Property

Public Sub New()

End Sub

End Class

A continuación instanciaremos la clase y con ello, inicializaremos adicionalmente la
propiedad que hemos declarado.

El código quedará de la siguiente forma:

Dim claseCliente As New Cliente() With {.Nombre = "Pedro"}

Esta forma de inicializar objetos nos permite ser más productivos y prácticos y
ahorrar tiempo y líneas de código en nuestras aplicaciones.

Como podemos apreciar, la forma de inicializar las propiedades o las variables de
una clase es utilizando la palabra claveWithseguida de llaves, y dentro de las
llaves, unpunto seguido de la propiedad o variable que queremos inicializar con
sus valor de inicialización. Si tenemos más de una variable a inicializar, deberemos
separarlas por comas.

Intellisensenos ayudará enormemente a inicializar la propiedad o variable que
deseemos, ya que vincula de forma directa el código de la clase haciéndonos más
fácil el trabajo con los miembros de la clase.

Ver vídeo de esta lección(Clases 5: Constructores)-video en
Visual Studio 2005 válido para Visual Studio 2008



Estructuras: Tipos por valor definidos por el usuario
De la misma forma que podemos definir nuestros propios tipos de datos por
referencia, Visual Basic 2008 nos permite crear nuestros propios tipos por valor.
Para crear nuestros tipos de datos por referencia, usamos la "instrucción"Class,
por tanto es de esperar que también exista una instrucción para crear nuestros
tipos por valor, y esa instrucción es:Structure, por eso en Visual Basic 2008 a los
tipos por valor definidos por el usuario se llaman estructuras.

Las estructuras pueden contener los mismos miembros que las clases, aunque
algunos de ellos se comporten de forma diferente o al menos tengan algunas
restricciones, como que los campos definidos en las estructuras no se pueden
inicializar al mismo tiempo que se declaran o no pueden contener constructores
"simples", ya que el propio compilador siempre se encarga de crearlo, para así
poder inicializar todos los campos definidos.

Otra de las características de las estructuras es que no es necesario crear una
instancia para poder usarlas, ya que esun tipo por valor y los tipos por valor no
necesitan ser instanciados para que existan.

Definir una estructura

Las estructuras se definen usando la palabraStructureseguida del nombre y
acaba usando las instruccionesEnd Structure.

El siguiente código define una estructura llamada Punto en la que tenemos dos
campos públicos.

Structure Punto

Public X As Integer

Public Y As Integer

End Structure



Para usarla podemos hacer algo como esto:

Dim p As Punto

p.X = 100

p.Y = 75

También podemos usarNewal declarar el objeto:

Dim p As New Punto

Aunque en las estructuras, usarNew, sería algo redundante y por tanto no
necesario.

Las estructuras siempre se almacenan en la pila, por tanto deberíamos tener la
precaución de no crear estructuras con muchos campos ocon muchos miembros,
ya que esto implicaría un mayor consumo del "preciado" espacio de la pila.

Constructores de las estructuras

Tal y como hemos comentado, las estructuras siempre definen un constructor sin
parámetros, este constructor no lo podemos definir nosotros, es decir, siempre
existe y es el que el propio compilador de Visual Basic 2008 define.
Por tanto, si queremos agregar algún constructor a una estructura, este debe
tener parámetros y, tal como ocurre con cualquier método o como ocurre en las
clases, podemos tener varias sobrecargas de constructores parametrizados en las
estructuras. La forma de definir esos constructores es como vimos en las clases:
usando distintas sobrecargas de un método llamadoNew, en el caso de las
estructuras también podemos usar la palabra claveMe para referirnos a la
instancia actual.

Esto es particularmente práctico cuando los parámetros del constructor se llaman
de la misma forma que los campos declarados en la estructura, tal como ocurre en
el constructor mostrado en la siguiente definición de la estructura Punto.

Structure Punto

Public X As Integer

Public Y As Integer

'

Sub New(ByVal x As Integer, ByVal y As Integer)

Me.X = x

Me.Y = y

End Sub

End Structure

Nota:
Tanto en las estructuras como en las clases podemos tener



constructores compartidos, (Shared), en el caso de las estructuras,
este tipo de constructor es el único que podemos declarar sin
parámetros.

Destructores de las estructuras

Debido a que las estructuras son tipos porvalor y por tanto una variable declarada
con un tipo por valor "contiene el valor en si misma", no podemos destruir este
tipo de datos, lo más que conseguiríamos al asignarle un valor nulo (Nothing) sería
eliminar el contenido de la variable, pero nunca podemos destruir ese valor. Por
tanto, en las estructuras no podemos definir destructores.

Los miembros de una estructura

Como hemos comentado, los miembros o elementos que podemos definir en una
estructura son los mismos que ya vimos enlas clases. Por tanto aquí veremos las
diferencias que existen al usarlos en las estructuras.

Campos

Como vimos, las variables declaradas a niveldel tipo, son los campos, la principal
diferencia con respecto a las clases, es que los campos de una estructura no
pueden inicialiarse en la declaración y el valor que tendrán inicialmente es un valor
"nulo", que en el caso de los campos de tipo numéricoses un cero.
Por tanto, si necesitamos que los campos tengan algún valor inicial antes de
usarlos, deberíamos indicarlo a los usuarios de nuestra estructura y proveer un
constructor que realice las inicializaciones correspondientes, pero debemos
recordar que ese constructor debe tener algún parámetro, ya que el
predeterminado sin parámetros no podemos "reescribirlo".

Los únicos campos que podemos inicializar al declararlos son los campos
compartidos, pero como tendremos oportunidad de ver, estos campos serán
accesibles por cualquier variable declarada y cualquier cambio que realicemos en
ellos se verá reflejado en el resto de "instancias" de nuestro tipo.

M‡todos y otros elementos

El resto de los miembros de una estructura se declaran y usan de la misma forma
que en las clases, si bien debemos tener en cuenta que el modificador de
accesibilidad predeterminado para los miembros de una estructura esPublic,
(incluso si son campos declarados conD¡m).

Otro detalle a tener en cuenta es que en una estructura siempre debe existir al
menos un evento o un campo no compartido, no se permiten estructuras en las
que solo tienen constantes, métodos y/o propiedades, estén o no compartidos.

Cómo usar las estructuras

Tal como hemos comentado las estructuras son tipos por valor, para usar los tipos
por valor no es necesario instanciarlos explícitamente, ya que el mero hecho de
declararlos indica que estamos creando un nuevo objeto en memoria. Por tanto, a
diferencia de las clases o tipos por referencia, cada variable definidacomo un tipo
de estructura será independiente de otras variables declaradas, aunque no las
hayamos instanciado.



Esta característica de las estructuras nos permite hacer copias "reales" no copia de
la referencia (o puntero) al objeto en memoria como ocurrecon los tipos por
referencia.
Veámoslos con un ejemplo.

Dim p As New Punto(100, 75)

Dim p1 As Punto

p1 = p

p1.X = 200

' p.X vale 100 y p1.X vale 200

En este trozo de código definimos einstanciamosuna variable del tipoPunto, a
continuación declaramos otra variable del mismo tipo y le asignamos la primera, si
estos tipos fuesen por referencia, tanto una como la otra estarían haciendo
referencia al mismo objeto en memoria, y cualquier cambio realizado a cualquiera
de las dos variables afectarían al mismo objeto, pero en el caso de las estructuras
(y de los tipos por valor), cada cambio que realicemos se hará sobre unobjeto
diferente, por tanto la asignación del valor200al campoX de la variablep1solo
afecta a esa variable, dejando intacto el valor original de la variablep.

Ver vídeo de esta lección(Estructuras)-video en Visual Studio
2005 válido para Visual Studio 2008



Accesibilidad y «mbito
Tal ycomo comentamos anteriormente, dependiendo de dónde y cómo estén
declarados los tipos de datos y los miembros definidos en ellos,tendremos o no
acceso a esos elementos.

Recordemos que el ámbito es el alcance con el que podemos acceder a un
elemento y depende de dónde esté declarado, por otro lado, la accesibilidad
depende de cómo declaremos cada uno de esos elementos.

Ámbito

Dependiendo de donde declaremos un miembro o un tipo, éste tendrá mayor
alcance o cobertura, o lo que es lo mismo, dependiendo del ámbito en el que
usemos un elemento, podremos acceder a él desde otros puntos de nuestro
código.
A continuación veremos con detallelos ámbitos en los que podemos declarar los
distintos elementos de Visual Basic 2008.

 Ámbito de bloque:Disponible únicamente en el bloque de código en el que
se ha declarado.
Por ejemplo, si declaramos una variable dentro de un bucleForo unIf Then,
esavariable solo estará accesible dentro de ese bloque de código.

 Ámbito de procedimiento:Disponible únicamente dentro del
procedimiento en el que se ha declarado. Cualquier variable declarada
dentro de un procedimiento (método o propiedad) solo estará accesible en
ese procedimiento y en cualquiera de los bloques internos a ese
procedimiento.

 Ámbito de módulo:Disponible en todo el código del módulo, la clase o la
estructura donde se ha declarado. Las variables con ámbito a nivel de
módulo, también estarándisponibles en los procedimientos declarados en el
módulo (clase o estructura) y por extensión a cualquier bloque dentro de
cada procedimiento.

 Ámbito de espacio de nombres:Disponible en todo el código del espacio
de nombres. Este es el nivel mayor de cobertura o alcance, aunque en este




